

# INITIATION AUX ALGORITHMES – LES BOUCLES LE LANGAGE VISUAL BASIC APPLICATIONS D'EXCEL (VBA EXCEL)

Les boucles sont les structures les plus puissantes de la programmation. On dit aussi STRUCTURES REPETITIVES. On utilise les instructions de boucle pour que le programme exécute un bloc d'instructions plusieurs fois.

## I – La structure WHILE WEND

|                 |              |   |
|-----------------|--------------|---|
| WHILE Condition | TANT QUE     | Tant que la condition est VRAI, les instructions vont être exécutées. |
| Instructions    |              |   |
| WEND            | FIN TANT QUE |   |

### Exemple 1 : On désire exécuter le calcul du montant TTC tant que l'utilisateur répond à O

```

Sub Exemple_1()
    Dim MontantHT As Single
    Const TxTVA As Single = 0.2
    Dim MontantTTC As Single
    Dim Reponse As String

    Reponse = "O"

    While Reponse = "O"
        MontantHT = InputBox("Entrer le montant HT")
        MontantTTC = MontantHT * (1 + TxTVA)
        MsgBox ("Le montant TTC est de : " & MontantTTC)

        Reponse = InputBox("Y a t-il un autre produit (O/N) ? ")

    Wend

    MsgBox "Fin du programme "
End Sub

```

Diagramme de la boucle WHILE WEND :

- Condition** : `Reponse = "O"` (TANT QUE Reponse est égal à O)
- Initialisation** : `Reponse = "O"` ('Stocke la réponse de l'utilisateur, O/N')
- Bloc d'instructions** : Calcul du montant TTC et affichage du message.
- Reévaluation** : `Reponse = InputBox("Y a t-il un autre produit (O/N) ? ")` (L'utilisateur saisit O ou N)
- Fin** : `MsgBox "Fin du programme "`

#### Remarques :

- Cette instruction est très importante : `Reponse = InputBox("Y a t-il un autre produit (O/N) ? ")` car elle permet de sortir de la boucle
- Si elle est absente, on aurait une boucle infinie (qui va se répéter indéfiniment)

### Exemple 2 :

Réaliser un programme qui invite l'utilisateur à saisir un mot de passe. Si le mot de passe saisi est « test », l'ordinateur affichera le message « bienvenue ».

```

Sub Exemple_2()
    Dim MotDePasse As String
    Dim MotSaisi As String
    MotDePasse = "test"

    MotSaisi = InputBox("Saisir votre mot de passe ")

    While MotSaisi <> MotDePasse
        MsgBox "Mot de passe incorrect"
        MotSaisi = InputBox("Saisir votre mot de passe ")
    Wend

    MsgBox "Bienvenue"
End Sub

```

Diagramme de la boucle WHILE WEND :

- Le bon mot de passe** : `MotDePasse = "test"`
- Saisie du mot de passe saisi** : `MotSaisi = InputBox("Saisir votre mot de passe ")`
- Condition** : `MotSaisi <> MotDePasse` (TANT QUE le mot Saisi est différent du mot de passe)
- Bloc d'instructions** : Affichage du message d'erreur et saisie du mot de passe.
- Fin** : `MsgBox "Bienvenue"`

### Exemple 3 :

Modification du programme précédent. Si l'utilisateur saisit 3 fois un mot de passe incorrect, le programme affichera « Nombre d'essai maximal atteint ». Si le mot de passe saisi est « test », l'ordinateur affichera le message « bienvenue ».

```

Sub Exemple_3()
    Dim MotDePasse As String
    Dim MotSaisi As String
    MotDePasse = "test"
    Dim i As Integer

    MotSaisi = InputBox("Saisir votre mot de passe ")
    i = 0

    While MotSaisi <> MotDePasse
        i = i + 1

        If i >= 3 Then
            MsgBox "Nombre d'essai maximal atteint"
            Exit Sub
        End If

        MotSaisi = InputBox("Saisir votre mot de passe ")
    Wend

    MsgBox "Bienvenue"

End Sub

```

Variable qui va compter le nombre de fois ou un mot de passe incorrect sera saisi. On appelle cela un compteur

Initialisation. Au départ, aucun mot de passe incorrect saisi.  $i = 0$

Tant que le mot de passe saisi sera différent du bon mot de passe,  $i$  va augmenter de 1 (INCREMENTER)

Si  $i \geq 3$ , on affiche un message et on quitte le programme

Cette instruction est très importante car il faut demander à l'utilisateur de saisir un autre mot de passe.

#### Exemple 4: Calcul de la prime d'ancienneté

On désire calculer la prime d'ancienneté tant qu'il y aura des salariés.

| Ancienneté    | Prime          |
|---------------|----------------|
| >10           | 10% du salaire |
| Entre 5 et 10 | 5% du salaire  |
| Sinon         | 2% du salaire  |

```

Sub Exemple_4()
    Dim NomSalarie As String
    Dim Anciennete As Single
    Dim Salaire As Single
    Dim PrimeAnciennete As Single
    Dim Reponse As String

    Reponse = "O"

    While Reponse = "O"
        NomSalarie = InputBox("Entrer le nom du salarié")
        Anciennete = InputBox("Entrer l'ancienneté")
        Salaire = InputBox("Entrer le salaire")

        If Anciennete > 10 Then
            PrimeAnciennete = Salaire * 0.1
        Else
            If Anciennete >= 5 Then
                PrimeAnciennete = Salaire * 0.05
            Else
                PrimeAnciennete = Salaire * 0.02
            End If
        End If

        MsgBox "La prime d'ancienneté du salarié : " & NomSalarie & " est de : " & PrimeAnciennete

        Reponse = InputBox("Y a t-il un autre salarié ? ")
    Wend

    MsgBox "Terminé"

End Sub

```

1. Quelle est la condition qui permet d'entrer dans la boucle ? \_\_\_\_\_
2. Encadrer le bloc d'instructions de la boucle
3. Quel est le rôle de la variable Reponse ? \_\_\_\_\_
4. Cette instruction est-elle importante ? Reponse = InputBox("Y a t-il un autre salarié ? ")  
Pourquoi ?

### Exemple 5: Calcul de la prime d'ancienneté

On désire calculer la prime d'ancienneté tant qu'il y aura des salariés. On aimerait que le programme affiche à la fin :

- Le nombre de salariés saisi
- Le total de la prime d'ancienneté pour tous les salariés

| NbSalarie   | 0          | TotalPrimeAnciennete | 0                |           |                      |
|-------------|------------|----------------------|------------------|-----------|----------------------|
| Nom Salarié | Ancienneté | Salaire              | Prime anciennete | NbSalarie | TotalPrimeAnciennete |
| Fred        | 5          | 2500                 | 125              |           |                      |
| Marc        | 12         | 1500                 | 150              |           |                      |
| Bon         | 2          | 1450                 | 29               |           |                      |

```
Sub Exemple_5()  
    Dim NomSalarie As String  
    Dim Anciennete As Single  
    Dim Salaire As Single  
    Dim PrimeAnciennete As Single  
    Dim Reponse As String  
  
    _____  
    _____  
  
    Reponse = "O"  
    _____  
    _____  
  
    While Reponse = "O"  
  
        NomSalarie = InputBox("Entrer le nom du salarié")  
        Anciennete = InputBox("Entrer l'ancienneté")  
        Salaire = InputBox("Entrer le salaire")  
  
        If Anciennete > 10 Then  
            PrimeAnciennete = Salaire * 0.1  
        Else  
            If Anciennete >= 5 Then  
                PrimeAnciennete = Salaire * 0.05  
            Else  
                PrimeAnciennete = Salaire * 0.02  
            End If  
        End If  
  
        MsgBox "La prime d'ancienneté du salarié : " & NomSalarie & " est de : " & PrimeAnciennete  
        _____  
        Reponse = InputBox("Y a t-il un autre salarié ? ")  
    Wend  
  
    MsgBox "Le nombre de salariés saisi : " & _____  
    MsgBox "Le total de la prime d'ancienneté est de : " & _____  
    MsgBox "Terminé"  
  
End Sub
```

## II – La structure FOR - NEXT

La boucle **For ... Next** permet de répéter des instructions un nombre précis de fois c-à-d qu'on connaît exactement combien de fois la boucle va s'exécuter.

|   |  |
|---|--|
| For compteur = ValeurDepart To ValeurFin<br>[instructions]<br>Next compteur | Pour compteur = ValeurDepart jusqu'à ValeurFin<br>[instructions]<br>Fin pour |
|---|--|

### Exemple 1 : Calcul de la moyenne des salaires

On saisit le nombre de salariés ; Pour chaque salarié, on va saisir le salaire, calculer le total des salaires et ensuite faire la moyenne.

```

Sub Exemple_7()
    Dim Nombre As Integer      'Nombre de salaries
    Dim Salaire As Single
    Dim i As Integer          'Compteur
    Dim TotalSalaire As Single

    TotalSalaire = 0

    Nombre = InputBox("Entrer le nombre de salariés")

    For i = 1 To Nombre
        Salaire = InputBox("Entrer le montant du salaire")
        TotalSalaire = TotalSalaire + Salaire
    Next
    MsgBox "La moyenne des salaires versés est de : " & TotalSalaire / Nombre
End Sub

```

SI Nombre = 3, les instructions de la boucle s'exécuteront 3 fois.

Instructions de la boucle.

### Exemple 2 : Calcul de la moyenne des notes

On saisit le nombre de notes ;on aimerait connaitre la moyenne et le nombre de notes supérieur ou égal à 10

```

Sub Exemple_8()
    Dim Nombre As Integer      'Nombre de notes
    Dim Note As Single
    Dim NbNotesSup10 As Integer
    Dim SommeNotes As Single
    Dim i As Integer

    SommeNotes = 0

    Nombre = InputBox("Entrer le nombre de notes")

    For i = 1 To Nombre
        Note = InputBox("Entrer la note")
        While Note < 0 Or Note > 20
            MsgBox "La note doit être comprise entre 0 et 20"
            Note = InputBox("Saisir la note")
        Wend
        If Note >= 10 Then
            NbNotesSup10 = NbNotesSup10 + 1
        End If
        SommeNotes = SommeNotes + Note
    Next

    MsgBox "La moyenne est de : " & SommeNotes / Nombre
    MsgBox "Le nombre de note supérieur ou égal à 10 : " & NbNotesSup10
End Sub

```